

Формулы и функции для финансовых расчетов

В ЭТОЙ ГЛАВЕ...

Стоимость денег во времени

Кредитный калькулятор

Расчет параметров инвестиций

Расчет амортизации

Не секрет, что чаще всего Excel используется для всевозможных вычислений, связанных с финансовыми расчетами. Каждый день люди принимают тысячи решений, основываясь на результатах вычислений в электронных таблицах. Принимаемые решения могут быть как очень простыми (*могу ли я купить этот автомобиль?*), так и достаточно сложными (*окупятся ли инвестиции в компанию ABC за 18 месяцев?*). В настоящей главе описаны основные приемы, применяемые в Excel для финансовых расчетов.

Стоимость денег во времени

Номинальная стоимость денег не всегда может соответствовать фактической. Ключевое понятие здесь — *стоимость денег во времени (СДВ)*. Вычисления, выполняемые над денежными суммами, могут производиться в прошлом, настоящем или в будущем времени. Выбор типа вычислений основывается на том, что сумма денег увеличивается, поскольку постоянно поступают начисления по процентам. Другими словами, вложив один рубль сегодня, завтра вы получите больше.

Например, предположим, что богатенький заокеанский дядюшка решил подарить вам деньги, причем вы должны выбрать один из перечисленных ниже вариантов.

- Получить 800 000 руб. сегодня.
- Получить 950 000 руб. через год.
- Получить 1 200 000 руб. через пять лет.
- Получать 15 000 руб. каждый месяц на протяжении пяти лет.

Чтобы получить максимальную выгоду, нужно учитывать не только собственную номинальную сумму, но и ту *стоимость*, которую будут представлять деньги на момент, когда вы их получите на руки.

Стоимость денег во времени зависит от ваших возможностей и перспектив. Другими словами, вы можете быть либо кредитором, либо заемщиком. Когда вы берете кредит на покупку автомобиля, вы выступаете в роли заемщика, а учреждение, которое дает вам деньги, — в роли кредитора. Когда же вы вкладываете деньги в банк на депозит (сберегательный счет), вы — кредитор, поскольку занимаете свои деньги банку, а банк является вашим заемщиком.

Со *стоимостью денег во времени* связано еще несколько понятий.


- **Приведенная (текущая) стоимость.** Это *основная* сумма, или *капитал*, на который начисляются проценты. Например, если на депозит в банке вкладывается 500 000 руб., то эта величина представляет собой капитал или приведенную (текущую) стоимость вложенных денег. Если берется ссуда размером 1 500 000 руб. на приобретение автомобиля, то данная сумма будет основной или приведенной стоимостью ссуды. Приведенная стоимость может быть как положительной, так и отрицательной.
- **Будущая стоимость.** Представляет собой сумму приведенной (текущей) стоимости и начисленным по ней процентам. Например, если на депозитный счет в банке вкладывается 50 000 руб. на пять лет под 6% годовых, то в конце срока можно будет получить 66 911,28 руб. Последняя сумма будет будущей стоимостью инвестиции. Если же берется ссуда на три года на покупку автомобиля в размере 150 000 руб. под 7% годовых, то в конце срока нужно будет выплатить 240 867,22 руб. Иными словами, нужно будет вернуть основную сумму плюс проценты. Будущая стоимость, в зависимости от перспективы (кредитор или заемщик), может быть положительной или отрицательной.
- **Взнос.** Взносом (платежом) может быть либо капитал, либо капитал и начисленные на него проценты. Если каждый месяц вкладывается по 1 000 руб. на депозитный счет, то 1 000 руб. — это взнос. Если для погашения ссуды ежемесячный взнос составляет 8 250 руб., то он состоит из тела кредита (основной суммы) и начисленных процентов.
- **Процентная ставка.** Часть основной суммы (в процентах), начисляемая за определенный период (как правило, за год). Например, деньги могут быть вложены в банк на депозит с процентной ставкой 5,5% годовых. Или процентная ставка по кредиту может составлять 7,75% в год.
- **Период.** Промежуток времени, по истечении которого выплачиваются проценты (например, банк может выплачивать проценты по депозитам ежеквартально, тогда как кредит за купленный автомобиль чаще всего приходится гасить ежемесячно).
- **Срок.** Промежуток времени, на который вкладываются или берутся в кредит деньги. Например, деньги вкладываются в банк на депозит на срок один год, или ссуда берется на срок 30 лет.

Кредитный калькулятор

Теперь рассмотрим, как производится расчет выплат по кредиту. Следует помнить, что кредит состоит из перечисленных ниже компонентов.

- Тело кредита (собственно размер ссуды, которую вы взяли в банке)
- Процентная ставка
- Сроки оплаты
- Размер регулярного платежа

Если известны любые три параметра, то можно создать формулу для вычисления четвертого параметра.



Все вычисления в этом разделе производятся для ссуды с фиксированными процентной ставкой и сроком.

Функции для расчета выплат по кредиту

В настоящем разделе описывается шесть функций: ПЛТ, ОСПЛТ, ПРПЛТ, СТАВКА, КПЕР и ПС. В табл. 15.1 описаны аргументы этих функций.

Таблица 15.1. Аргументы функций для финансовых расчетов

Аргумент	Описание
<i>Ставка</i>	Процентная ставка за один период. Если она выражена в процентах за год, то эту величину нужно разделить на количество периодов (месяцев)
<i>Кпер</i>	Общее количество выплат (периодов)
<i>Период</i>	Заданный период, который должен быть меньше или равен значению параметра <i>Кпер</i>
<i>Пс</i>	Приведенная стоимость, или основная сумма
<i>Плт</i>	Сумма, которая вносится за каждый период (постоянная величина)
<i>Бс</i>	Необязательный аргумент, равный будущей стоимости после последней выплаты. Если аргумент опущен, то он полагается равным нулю. (Например, будущая стоимость ссуды после ее погашения, очевидно, равна нулю.)
<i>Тип</i>	Указывает, когда должны производиться платежи. Равен нулю, если платежи выполняются в конце периода, и единице — если в начале. Это необязательный аргумент, который по умолчанию равен нулю
<i>Оценка</i>	Используется в функции СТАВКА и означает начальную оценку ожидаемого результата. Функция СТАВКА выполняет пошаговое вычисление. Поэтому, если полученный результат не будет сходиться, следует изменить значение этого аргумента

Функция ПЛТ

Функция ПЛТ позволяет рассчитать размер выплат по кредиту (основная сумма плюс проценты) за один период, полагая постоянными размер выплат и процентную ставку. Эта функция имеет такой синтаксис:

ПЛТ (*Ставка*; *Кпер*; *Пс*; *Бс*; *Тип*)

Следующая формула возвращает размер ежемесячных выплат по кредиту, размер которого 50 000 руб., а процентная ставка составляет 6% годовых. Кредит взят на срок четыре года (48 месяцев).

=ПЛТ (6%/12; 48; -50000)

Вычислив формулу, получим, что размер ежемесячных выплат по ссуде равен 1 174,25 руб. В качестве первого аргумента *Ставка* указан размер годовой ставки по кредиту, деленный на 12 месяцев (количество месяцев в году). Обратите также внимание на то, что третий аргумент (*Пс*, приведенная стоимость) отрицательный, — это указывает на то, что деньги были взяты в долг.

Функция ОСПЛТ


Функция **ОСПЛТ** возвращает основную часть платежа по кредиту за определенный период, подразумевая постоянным размер выплат и фиксированную процентную ставку. Функция имеет следующий синтаксис:

ОСПЛТ (*Ставка*; *Период*; *Кпер*; *Пс*; *Бс*; *Тип*)

Следующая формула возвращает основную часть выплаты за первый месяц по ссуде 50 000 руб., взятой под 6% годовых. Ссуда взята на срок четыре года (48 месяцев).

=ОСПЛТ (6%/12; 1; 48; -50000)

Вычислив формулу, получим, что основная часть первой выплаты равна 924,25 руб., или приблизительно 78,7% от общей суммы выплаты. Если в качестве второго аргумента ввести число 48 (для вычисления основной части последней выплаты), то формула вернет 1 168,41 руб., т.е. приблизительно 99,5% от общей суммы выплаты.



Чтобы вычислить размер основного кумулятивного платежа, который нужно будет сделать между любыми двумя определенными периодами, можно воспользоваться функцией **ОБЩДОХОД**. Функция содержит два дополнительных аргумента: *нач_период* и *кон_период*. В версиях, предшествовавших Excel 2007, функция **ОБЩДОХОД** становилась доступной только после установки надстройки *Пакет анализа*.

Функция ПРПЛТ


Функция **ПРПЛТ** вычисляет ту часть общей суммы выплат по кредиту, которая идет на погашение процентов, полагая постоянными размер выплат и процентную ставку. Функция имеет следующий синтаксис:

ПРПЛТ (*Ставка*; *Период*; *Кпер*; *Пс*; *Бс*; *Тип*)

Формула, приведенная ниже, вычисляет размер первой выплаты по процентам для ссуды 50 000 руб., взятой под 6% годовых. Ссуда взята на 4 года (48 месяцев).

=ПРПЛТ (6%/12; 1; 48; -50000)

После вычисления формулы получим, что сумма выплат по процентам за первый месяц составляет 250,00 руб. Но размер выплат по процентам за последний месяц будет всего 5,84 руб.



Чтобы вычислить размер кумулятивного платежа по процентам, который нужно будет сделать между любыми двумя определенными периодами, следует воспользоваться функцией ОБЩПЛАТ. Функция содержит два дополнительных аргумента: *нач_период* и *кон_период*. В версиях, предшествовавших Excel 2007, функция ОБЩПЛАТ становилась доступной только после установки надстройки Пакет анализа.

Функция СТАВКА

Функция СТАВКА позволяет вычислить размер процентной ставки по кредиту за один период на основании указанного общего количества периодов выплат, размера регулярного платежа и размера ссуды. Функция СТАВКА имеет следующий синтаксис:

СТАВКА (Кпер; Плт; Пс; Бс; Тип; Оценка)

Следующая формула вычисляет годовую процентную ставку для ссуды размером 50 000 руб., взятой на 48 месяцев. Размер регулярного платежа составляет 1 174,25 руб.

=СТАВКА(48;1174,25;-50000)*12

Формула вернет 6,00%. Заметьте, что результат вычисления функции умножается на 12, поскольку функция возвращает процентную ставку за один период (месяц). Поэтому, чтобы получить годовую процентную ставку, нужно умножить полученное значение на количество периодов выплат в году (в нашем примере — на 12).

Функция КПЕР

Функция КПЕР возвращает общее количество периодов выплат по ссуде по заданным размеру ссуды, процентной ставке и размеру регулярного платежа. Функция имеет следующий синтаксис:

КПЕР (Ставка; Плт; Пс; Бс; Тип)

Формула, приведенная ниже, вычисляет количество выплат по ссуде размером 50 000 руб. и размеру ежемесячной выплаты, равному 1 174,25 руб. Ссуда взята под 6% годовых.

=КПЕР(6%/12;1174,25;-50000)

С небольшой погрешностью (меньше 0,0001) функция вернет число 48. Это значит, что всего нужно будет сделать 48 (за 48 месяцев) выплат по кредиту. Полученный результат оказался неточным из-за того, что размер выплат указан с точностью до одной копейки, т.е. округлен.

Функция ПС

Функция ПС вычисляет приведенную стоимость ссуды (т.е. первоначальный размер кредита) по заданным процентной ставке, количеству периодов и размеру регулярного платежа. Функция имеет такой синтаксис:

ПС (Ставка; Кпер; Плт; Бс; Тип)

Следующая формула возвращает размер первоначальной ссуды, взятой на 48 месяцев под 6% годовых. Размер регулярного платежа равен 1 174,25 руб. в месяц.

=ПС(6%/12;48;-1174,25)

Формула вернет 49 999,94 руб. Поскольку размер регулярного платежа указан с точностью до одной копейки, результат имеет погрешность в 6 коп.

Пример кредитного калькулятора

На рис. 15.1 показан рабочий лист, используя который можно вычислить размер регулярного платежа по кредиту. В ячейке B1 находится сумма кредита, а в ячейке B2 — годовая процентная ставка. В ячейке B3 содержится длительность одного периода, выраженная в месяцах. Например, если в ячейке B3 находится число 1, то это значит, что выплаты производятся ежемесячно. Общее количество периодов хранится в ячейке B4. В примере, показанном на рис. 15.1, вычисляется размер ежемесячной выплаты по кредиту размером 100 000 руб., взятым под 9,5% годовых на 36 месяцев. В ячейке B6 содержится следующая формула:

=ПЛТ (B2*(B3/12);B4;-B1)

	A	B	C
1	Сумма кредита:	100 000,00 р.	
2	Годовая процентная ставка:	9,50%	
3	Периодичность платежей (мес.):	1	
4	Количество периодов:	36	
5			
6	Регулярный платеж:	3 203,29 р.	
7			
8			
9	Период:	36	
10	Тело:	3 178,13 р.	
11	Проценты:	25,16 р.	
12			
13			

Рис. 15.1. Вычисление размера регулярного платежа по кредиту с помощью функции ПЛТ

Обратите внимание на то, что первым аргументом является выражение, которое вычисляет *периодическую процентную ставку* на основе значений годовой процентной ставки и периодичности выплат. Поэтому если выплаты производятся ежеквартально на протяжении трех лет, то периодичность выплат будет равна 3, а количество периодов — 12. Для того чтобы вычислить периодическую процентную ставку, нужно умножить годовую ставку на 3/12.



Рабочую книгу с примерами для данного раздела можно найти в файле `loan payment.xlsx` на сайте книги. Локализованный файл называется `Кредитный калькулятор.xlsm`.

В ячейках B10 и B11 вычисляются размеры платежей, погашающих тело кредита и проценты за период, указанный в ячейке B9. Не забывайте, что значение в ячейке B9 должно быть меньше или равно значению ячейки B4. Иными словами, заданный период не должен превышать общее количество периодов.

Формула в ячейке B10 представлена ниже. Она позволяет вычислить размер тела кредита для заданного в ячейке B9 периода.

=ОСПЛТ (B2*(B3/12);B9;B4;-B1)

Приведенная ниже формула (ячейка В11) вычисляет размер выплат по процентам для заданного в ячейке В9 периода:

=ПРПЛТ (В2* (В3/12) ;В9;В4; -В1)

Обратите внимание на то, что сумма ячеек В10 и В11 всегда равна размеру регулярного платежа по кредиту, который вычисляется в ячейке В6. Тем не менее соотношение между платежами по кредиту и выплатам по процентам зависит от текущего периода. (Доля платежей по кредиту с каждой последующей выплатой увеличивается, а выплат по процентам — уменьшается.) На рис. 15.2 графически показаны размеры выплачиваемых сумм для всего срока погашения кредита.

Расчеты по кредитной карточке

Предположим, у вас есть кредитная карточка с определенной начальной суммой и вы хотите узнать за сколько месяцев вы сможете погасить этот кредит при условии внесения минимально возможного ежемесячного платежа. На рис. 15.3 показан рабочий лист, на котором выполняются вычисления такого типа.




Рис. 15.2. Соотношение между платежами по кредиту и выплатам по процентам в зависимости от периода выплат

	A	B	C
1	Начальная сумма:	10 000,00 р.	
2	Годовая процентная ставка:	18,25%	
3	Минимальный процент платежа:	2,00%	
4	Минимальная сумма платежа:	200,00 р.	
5	Реальный ежемесячный платеж:	200	
6			
7	Количество платежей:	94,7	
8	Общая сумма:	18 932,93 р.	
9	Из них процентов:	8 932,93 р.	
10			
11			
12			
13			

Рис. 15.3. На этом рабочем листе подсчитывается количество платежей, требуемых для погашения баланса по кредитной карте при условии внесения минимального ежемесячного платежа

В диапазоне В1:В5 записаны исходные данные по кредитной карточке. В нашем примере начальная сумма на кредитной карточке составляет 10 000 руб., годовая процентная ставка равна 18,25%. Здесь принято, что минимальная сумма платежа составляет 2% от суммы, хранимой на кредитной карте, т.е. в данном случае минимальный платеж будет составлять 200 руб. в месяц. В ячейку В5 вы можете ввести любую другую сумму, но она должна быть достаточной для погашения всей суммы кредита. Например, вы можете вносить ежемесячно по 500 руб., чтобы погасить кредит быстрее. С другой стороны, очевидно, что 100 руб. в месяц будет недостаточно, поэтому формула возвратит ошибку.



Рабочую книгу с примерами для данного раздела можно найти в файле `credit card payments.xlsx` на сайте книги. Локализованный файл называется Платежи по кредитной карте.xlsx.

В диапазоне B7:B9 выполняются вычисления. Приведенная ниже формула, записанная в ячейке B7, вычисляет, за сколько месяцев вы сможете погасить баланс по кредитной карте.

=КПЕР(B2/12;B5;-B1;0)

Здесь предполагается, что в конце вычисленного периода баланс по карте должен быть погашен, поэтому последний аргумент в функции КПЕР равен нулю. Простая формула в ячейке B8 вычисляет общую сумму, которую вы должны будете заплатить за все время с учетом начисленных процентов.

=B7*B5

Формула в ячейке B9 вычисляет сумму, начисленную по процентам:

=B8-B1

В этом примере доходчиво показано, что кредит в 10 000 руб. вы будете гасить 95 месяцев (почти 8 лет!) при условии внесения ежемесячного минимального платежа в 200 руб. За это же время на взятые в кредит по карте 10 000 руб. вам будет начислено 8 932,93 руб. процентов. Конечно, эти вычисления предполагают, что за все время пользования карточкой вы не снимали с нее никаких дополнительных средств. Из этого примера вы уже должны были понять, почему банки так упорно рассылают кредитные карты по обычной почте буквально направо и налево, а также всячески поощряют вас пользоваться такими картами.

На рис. 15.4 показаны дополнительные вычисления, связанные с нашим примером с кредитной карточкой. Например, если вы захотите погасить баланс по карточке за 12 месяцев, то каждый месяц вы должны будете вносить по 917,99 руб. При этом общая сумма выплат составит 11 015,88 руб., а начисления по процентам — 1 015,88 руб. В ячейке B13 записана формула

=ПЛТ(\$B\$2/12;A13;-\$B\$1)

11				
12	Периодичность выплат (мес.)	Платеж	Всего	Всего процентов
13	2	5 114,35р.	10 228,70р.	228,70р.
14	6	1 756,50р.	10 538,99р.	538,99р.
15	12	917,99р.	11 015,88р.	1 015,88р.
16	24	500,45р.	12 010,80р.	2 010,80р.
17	36	362,78р.	13 060,06р.	3 060,06р.
18	48	295,06р.	14 162,78р.	4 162,78р.
19	60	255,30р.	15 317,77р.	5 317,77р.
20	72	229,49р.	16 523,53р.	6 523,53р.
21	84	211,65р.	17 778,29р.	7 778,29р.
22	96	198,75р.	19 080,05р.	9 080,05р.
23	108	189,14р.	20 426,64р.	10 426,64р.
24	120	181,80р.	21 815,73р.	11 815,73р.
25	132	176,10р.	23 244,89р.	13 244,89р.
26				
27				
28				
29				

Рис. 15.4. В столбце B приведены размеры платежей, которые нужно сделать, чтобы погасить баланс по кредитной карте за указанный период

Создание графика амортизации кредита

График амортизации кредита представляет собой таблицу, содержащую различную информацию по каждому периоду выплаты. На рис. 15.5 показан рабочий лист, который содержит формулы, используемые для создания графика амортизации кредита.


	A	B	C	D	E	F	G	H
1	Сумма кредита:	100 000,00 p.						
2	Годовая процентная ставка:	7,25%						
3	Периодичность платежей (мес.):	1						
4	Число периодов:	36						
5								
6	График амортизации кредита							
7	Номер платежа	Размер платежа	Сумма выплат	Выплаты по процентам	Сумма по процентам	Тело кредита	Сумма по кредиту	Остаток
8								100 000,00 p.
9	1	3 099,15 p.	3 099,15 p.	604,17 p.	604,17 p.	2 494,99 p.	2 494,99 p.	97 505,01 p.
10	2	3 099,15 p.	6 198,31 p.	589,09 p.	1 193,26 p.	2 510,06 p.	5 005,05 p.	94 994,95 p.
11	3	3 099,15 p.	9 297,46 p.	573,93 p.	1 767,19 p.	2 525,23 p.	7 530,27 p.	92 469,73 p.
12	4	3 099,15 p.	12 396,61 p.	558,67 p.	2 325,86 p.	2 540,48 p.	10 070,75 p.	89 929,25 p.
13	5	3 099,15 p.	15 495,76 p.	543,32 p.	2 869,18 p.	2 555,83 p.	12 626,58 p.	87 373,42 p.
14	6	3 099,15 p.	18 594,92 p.	527,88 p.	3 397,06 p.	2 571,27 p.	15 197,86 p.	84 802,14 p.
15	7	3 099,15 p.	21 694,07 p.	512,35 p.	3 909,41 p.	2 586,81 p.	17 784,66 p.	82 215,34 p.
16	8	3 099,15 p.	24 793,22 p.	496,72 p.	4 406,13 p.	2 602,44 p.	20 387,10 p.	79 612,90 p.
17	9	3 099,15 p.	27 892,38 p.	480,99 p.	4 887,12 p.	2 618,16 p.	23 005,26 p.	76 994,74 p.
18	10	3 099,15 p.	30 991,53 p.	465,18 p.	5 352,30 p.	2 633,98 p.	25 639,23 p.	74 360,77 p.
19	11	3 099,15 p.	34 090,68 p.	449,26 p.	5 801,56 p.	2 649,89 p.	28 289,12 p.	71 710,88 p.
20	12	3 099,15 p.	37 189,84 p.	433,25 p.	6 234,81 p.	2 665,90 p.	30 955,02 p.	69 044,98 p.
21	13	3 099,15 p.	40 288,99 p.	417,15 p.	6 651,96 p.	2 682,01 p.	33 637,03 p.	66 362,97 p.
22	14	3 099,15 p.	43 388,14 p.	400,94 p.	7 052,90 p.	2 698,21 p.	36 335,24 p.	63 664,76 p.
23	15	3 099,15 p.	46 487,29 p.	384,64 p.	7 437,54 p.	2 714,51 p.	39 049,75 p.	60 950,25 p.
24	16	3 099,15 p.	49 586,45 p.	368,24 p.	7 805,79 p.	2 730,91 p.	41 780,66 p.	58 219,34 p.
25	17	3 099,15 p.	52 685,60 p.	351,74 p.	8 157,53 p.	2 747,41 p.	44 528,07 p.	55 471,93 p.
26	18	3 099,15 p.	55 784,75 p.	335,14 p.	8 492,67 p.	2 764,01 p.	47 292,08 p.	52 707,92 p.
27	19	3 099,15 p.	58 883,91 p.	318,44 p.	8 811,11 p.	2 780,71 p.	50 072,79 p.	49 927,21 p.
28	20	3 099,15 p.	61 983,06 p.	301,64 p.	9 112,76 p.	2 797,51 p.	52 870,30 p.	47 129,70 p.
29	21	3 099,15 p.	65 082,21 p.	284,74 p.	9 397,50 p.	2 814,41 p.	55 684,71 p.	44 315,29 p.
30	22	3 099,15 p.	68 181,36 p.	267,74 p.	9 665,24 p.	2 831,41 p.	58 516,13 p.	41 483,87 p.
31	23	3 099,15 p.	71 280,52 p.	250,63 p.	9 915,87 p.	2 848,52 p.	61 364,65 p.	38 635,35 p.

Рис. 15.5. График амортизации кредита



Рабочую книгу с примерами для данного раздела можно найти в файле `loan amortization schedule.xlsx` на сайте книги. Локализованный файл называется График амортизации кредита.xlsm.

Данные по кредиту введены в диапазон B1:B4. Они используются в формулах, содержащихся в ячейках, начиная с девятой строки. В табл. 15.2 представлены формулы из строки 9 графика выплат. Формулы были скопированы во все строки графика до 488 строки. Поэтому с помощью этого рабочего листа можно рассчитать график погашения ссуды продолжительностью до 480 периодов (40 лет).



Формулы в тех строках, которые не относятся к периодам выплат будут возвращать ошибку. Поэтому для сокрытия данных в этих строках используются средства условного форматирования.




Подробнее об условном форматировании можно узнать в главе 21.

Таблица 15.2. Формулы, используемые для создания графика выплат

Ячейка	Формула	Описание
A9	=A8+1	Возвращает номер платежа
B9	=ПЛТ (\$B\$2* (\$B\$3/12) ; \$B\$4; -\$B\$1)	Вычисляет размер периодического платежа
C9	=C8+B9	Возвращает общую (кумулятивную) сумму выплат
D9	=ПРПЛТ (\$B\$2* (\$B\$3/12) ; A9; \$B\$4; -\$B\$1)	Вычисляет процентную часть выплаты
E9	=E8+D9	Вычисляет кумулятивную сумму выплат по процентам
F9	=ОСПЛТ (\$B\$2* (\$B\$3/12) ; A9; \$B\$4; -\$B\$1)	Вычисляет размер тела кредита за одну выплату
G9	=G8+F9	Вычисляет кумулятивную сумму тела кредита
H9	=H8-F9	Вычисляет остаток суммы кредита в конце каждого периода

Анализ параметров ссуды с помощью таблиц данных

Таблицы данных являются, по-видимому, одним из самых мощных средств Excel. Следует принять во внимание, что таблицы данных — это не совсем то же, что таблица (созданная с помощью команды Вставка⇒Таблицы⇒Таблица). Таблица данных — это удобный способ для подведения итогов вычислений, зависящих от одной или двух “изменяющихся” ячеек. В этом примере, изменяя значения одного или двух параметров ссуды, можно проследить их влияние на размер платежа. В следующих разделах описано, как создавать таблицы данных с одним или двумя входами.



В предыдущих русскоязычных версиях Excel таблицы данных назывались *таблицами подстановки*. — Примеч. ред.



Таблицы данных более подробно описаны в главе 35.



Рабочую книгу с примерами использования таблиц данных с одним и двумя входами можно найти в файле `loan data tables.xlsx` на веб-сайте книги. Локализованный файл называется `Анализ данных с помощью таблицы данных.xlsm`.

Создание таблицы данных с одним входом

Таблица данных с одним входом позволяет выполнить произвольное количество расчетов, изменяя значения в одной ячейке.

На рис. 15.6 показана таблица (диапазон B10:I13), в которой вычисляются три параметра ссуды (размер периодического платежа, общая сумма выплат по кредиту и общая сумма выплат по процентам) при различных значениях процентной ставки, которая изменяется от 7,00 до 8,50% (всего семь значений). В данном примере значения подставляются в ячейку B2.

	A	B	C	D	E	F	G	H	I
1	Сумма кредита:	100 000,00 р.							
2	Годовая процентная ставка:	7,25%							
3	Периодичность платежей (мес.):	1							
4	Число периодов:	36							
5									
6	Размер платежа:	3 099,15 р.							
7	Сумма выплат:	111 569,51 р.							
8	Сумма процентов:	11 569,51 р.							
9									
10			7,00%	7,25%	7,50%	7,75%	8,00%	8,25%	8,50%
11	Размер платежа:	\$3 099	3 087,71 р.	3 099,15 р.	3 110,62 р.	3 122,12 р.	3 133,64 р.	3 145,18 р.	3 156,75 р.
12	Сумма выплат:	\$111 570	111 157,55 р.	111 569,51 р.	111 982,39 р.	112 396,19 р.	112 810,92 р.	113 226,56 р.	113 643,13 р.
13	Сумма процентов:	\$11 570	11 157,55 р.	11 569,51 р.	11 982,39 р.	12 396,19 р.	12 810,92 р.	13 226,56 р.	13 643,13 р.
14									

Рис. 15.6. Пример таблицы данных с одним входом для вычисления параметров ссуды для различных значений процентной ставки

Чтобы создать таблицу данных с одним входом, выполните следующие действия.

- Введите формулы для вычисления параметров ссуды**, которые будут использоваться в таблице данных. В этом примере формулы находятся в диапазоне B6:B8.
- Введите значения изменяемой ячейки в соответствующие ячейки.** Здесь изменяемым значением является процентная ставка, значения которой находятся в ячейках C10:I10.
- В столбце, который находится слева от изменяемых значений**, введите ссылки на ячейки с формулами. В данном примере ссылки следует ввести в диапазон B11:B13. Например, в ячейке B11 содержится формула =B6.
- Выделите диапазон, который содержит все данные**, созданные в предыдущих пунктах. (В данном примере это диапазон B10:I13.)
- Выберите команду Данные⇒Работа с данными⇒Анализ «что если»⇒Таблица данных.** На экране появится диалоговое окно, показанное на рис. 15.7.
- В поле Подставлять значения по столбцам в укажите ссылку на изменяемую ячейку.** В данном случае изменяемое значение хранится в ячейке B2.
- Второе поле оставьте пустым.** Оно используется для создания таблиц данных с двумя входами.
- Щелкните на кнопке ОК.** Excel создаст формулу массива, которая использует функцию ТАБЛИЦА с одним аргументом.
- По желанию отформатируйте полученную таблицу.** Например, можно добавить фон для заголовков таблицы.




Рис. 15.7. Диалоговое окно Таблица данных

Обратите внимание на то, что формула массива была введена не во все ячейки выделенного диапазона. Первые строка и столбец диапазона, выбранного в п. 4, остались без изменений.



При создании таблицы данных в левом столбце (в п. 3 в нем мы указали ссылки на ячейки с формулами) содержатся результаты вычислений для заданного в ячейке B2 значения изменяемого параметра. В нашем примере эти значения повторяются в столбце D. Поэтому, чтобы избежать недоразумений, лучше всего скрыть значения в левом столбце. Для этого можно, например, сделать одинаковыми цвет шрифта и цвет фона для левого столбца.

Создание таблицы данных с двумя входами

Таблица данных с двумя входами содержит результаты вычислений по одной формуле при изменении двух входных параметров. На рис. 15.8 показана таблица данных с двумя входами (диапазон B10:I16), которая используется для вычисления размера платежа для семи значений процентной ставки и шести значений суммы кредита.

	A	B	C	D	E	F	G	H	I
1	Сумма кредита:	100 000,00 р.							
2	Годовая процентная ставка:	7,25%							
3	Периодичность платежей (мес.):	1							
4	Число периодов:	36							
5									
6	Размер платежа:	3 099,15 р.							
7	Сумма выплат:	111 569,51 р.							
8	Сумма процентов:	11 569,51 р.							
9									
10									
11									
12									
13									
14									
15									
16									

Рис. 15.8. Применение таблицы данных с двумя входами для вычисления размера платежа при разных параметрах ссуды

Чтобы создать таблицу данных с двумя входами, выполните такие действия.

- 1. Создайте формулу, которая будет вычислять данные, используемые в таблице подстановки.** В этом примере такая формула находится в ячейке B6. Формулы в ячейках B7 и B8 не используются.
- 2. Введите в строку значения первого изменяемого параметра.** В данном примере первым параметром является значение процентной ставки. Подставляемые значения хранятся в диапазоне C10:I10.
- 3. Введите в столбец значения второго изменяемого параметра.** В этом примере сумма кредита выступает в качестве второго параметра, значения которого содержатся в диапазоне B11:B16.
- 4. Создайте ссылку на ячейку, содержащую формулу для вычисления значений в таблице.** В нашем примере в ячейке B10 содержится ссылка =B6.
- 5. Выделите диапазон, содержащий все ячейки,** в которые были введены данные на предыдущих шагах. В данном случае — диапазон B10:I16.
- 6. Выберите команду Данные⇒Работа с данными⇒Анализ «что-если»⇒Таблица данных.** Excel отобразит одноименное диалоговое окно, показанное на рис. 15.7.

7. В поле Подставлять значения по столбцам в укажите ссылку на ячейку, которая содержит текущее значение первого параметра. В данном примере следует ввести ссылку на ячейку B2.
8. В поле Подставлять значения по строкам в введите ссылку на ячейку с текущим значением второго параметра таблицы подстановки. В нашем примере создайте ссылку на ячейку B1.
9. Щелкните на кнопке ОК. Excel вставит формулу массива с функцией ТАБЛИЦА с двумя аргументами.

После создания таблицы данных можно изменить формулу, по которой осуществляются расчеты. Для этого следует изменить ссылку в верхней левой ячейке таблицы подстановки. Например, если в этом примере заменить существующую ссылку в ячейке B10 на ссылку =B8, то в таблице будет рассчитываться общая сумма выплат по процентам, а если в ячейку B10 ввести ссылку =B7, то в таблице будет рассчитываться общая сумма выплат с учетом процентов.



Если таблица данных будет иметь очень большой размер, то скорость вычислений может существенно снизиться. Поэтому в Excel предусмотрен специальный режим вычислений для подобных случаев. Выберите команду Формулы⇒Вычисление⇒Параметры вычислений⇒Автоматически, кроме таблиц данных. После выбора этой команды пересчет таблиц данных будет выполняться не в автоматическом режиме, а после нажатия клавиши <F9>.

Расчет параметров погашения ссуды нерегулярными платежами

До сих пор во всех примерах из этой главы рассчитывались параметры погашения кредита (ссуды) регулярными платежами. Но в некоторых случаях ссуда погашается нерегулярно. Например, вы заняли денег друзьям без официального соглашения о том, как будут производиться выплаты. И если при этом вас будет интересовать, какую сумму вы получите в качестве процентов, то в таком случае нужно выполнить вычисления на основании дат реальных платежей.

На рис. 15.9 представлен рабочий лист, с помощью которого можно отслеживать состояние таких займов. Значение годовой процентной ставки ссуды вводится в ячейку B1, которой присвоено имя ПС. Размер и дата первоначальной ссуды находятся в ячейках B5 и C5 соответственно. Формулы, расположенные в 6-й строке и ниже, позволяют отследить состояние ссуды, а также производят дополнительные расчеты.

В столбце В содержатся размеры платежей, сделанных в день, указанный в столбце С. Заметьте, что выплаты производились нерегулярно. Также следует отметить, что в двух случаях (строки 11 и 24) размер выплаты отрицательный. Это значит, что производились дополнительные займы, которые добавлялись к остатку ссуды. Формулы в ячейках D и E вычисляют сумму выплат по процентам и основных выплат. Значения в столбцах F и G позволяют отследить общую сумму основных выплат и сумму выплат по процентам. Формулы в столбце H вычисляют остаток по ссуде после каждой выплаты. В табл. 15.3 приведены формулы, которые содержатся в строке 6, а также их краткое описание. Обратите внимание на то, что в каждой формуле используется функция ЕСЛИ, предназначенная для определения отсутствия даты платежа в столбце С. Если дата отсутствует, формула возвращает пустую строку и в ячейке таким образом данных не будет.

	A	B	C	D	E	F	G	H
1	Годовая процентная ставка (ПС):	6,50%						
2							История погашения ссуды	
3								
4		Размер платежа	Дата платежа	Проценты	Тело	Сумма платежей	Сумма по процентам	Баланс
5	Основная ссуда	-10 000,00 р.	08.06.2011					10 000,00 р.
6	1	200,00 р.	25.07.2011	83,70 р.	116,30 р.	200,00 р.	83,70 р.	9 883,70 р.
7	2	200,00 р.	09.08.2011	26,40 р.	173,60 р.	400,00 р.	110,10 р.	9 710,10 р.
8	3	200,00 р.	24.09.2011	79,54 р.	120,46 р.	600,00 р.	189,64 р.	9 589,64 р.
9	4	100,00 р.	09.12.2011	129,79 р.	-29,79 р.	700,00 р.	319,43 р.	9 619,43 р.
10	5	250,00 р.	19.01.2012	70,24 р.	179,76 р.	950,00 р.	389,67 р.	9 439,67 р.
11	Дополнительная ссуда	-500,00 р.	01.02.2012	21,85 р.	-521,85 р.	450,00 р.	411,52 р.	9 961,52 р.
12	6	200,00 р.	21.02.2012	35,48 р.	164,52 р.	650,00 р.	447,00 р.	9 797,00 р.
13	7	100,00 р.	27.02.2012	10,47 р.	89,53 р.	750,00 р.	457,47 р.	9 707,47 р.
14	8	1 000,00 р.	07.03.2012	15,56 р.	984,44 р.	1 750,00 р.	473,03 р.	8 723,03 р.
15	9	250,00 р.	22.03.2012	23,30 р.	226,70 р.	2 000,00 р.	496,33 р.	8 496,33 р.
16	10	200,00 р.	08.04.2012	25,72 р.	174,28 р.	2 200,00 р.	522,05 р.	8 322,05 р.
17	11	200,00 р.	25.04.2012	25,19 р.	174,81 р.	2 400,00 р.	547,24 р.	8 147,24 р.
18	12	1 000,00 р.	10.05.2012	21,76 р.	978,24 р.	3 400,00 р.	569,01 р.	7 169,01 р.
19	13	100,00 р.	22.05.2012	15,32 р.	84,68 р.	3 500,00 р.	584,33 р.	7 084,33 р.
20	14	200,00 р.	08.06.2012	21,45 р.	178,55 р.	3 700,00 р.	605,77 р.	6 905,77 р.
21	15	200,00 р.	25.06.2012	20,91 р.	179,09 р.	3 900,00 р.	626,68 р.	6 726,68 р.
22	16	100,00 р.	11.07.2012	19,17 р.	80,83 р.	4 000,00 р.	645,85 р.	6 645,85 р.
23	17	100,00 р.	21.07.2012	11,84 р.	88,16 р.	4 100,00 р.	657,68 р.	6 557,68 р.
24	Дополнительная ссуда	-500,00 р.	07.09.2012	56,05 р.	-556,05 р.	3 600,00 р.	713,74 р.	7 113,74 р.
25	18	100,00 р.	08.11.2012	78,54 р.	21,46 р.	3 700,00 р.	792,28 р.	7 092,28 р.
26	19	100,00 р.	23.11.2012	18,95 р.	81,05 р.	3 800,00 р.	811,23 р.	7 011,23 р.
27	20	200,00 р.	21.12.2012	34,96 р.	165,04 р.	4 000,00 р.	846,19 р.	6 846,19 р.
28	21	750,00 р.	11.01.2013	25,60 р.	724,40 р.	4 750,00 р.	871,79 р.	6 121,79 р.
29	22	750,00 р.	11.02.2013	33,80 р.	716,20 р.	5 500,00 р.	905,58 р.	5 405,58 р.
30	23	750,00 р.	11.03.2013	26,95 р.	723,05 р.	6 250,00 р.	932,54 р.	4 682,54 р.

Рис. 15.9. Рабочий лист, с помощью которого можно отследить параметры погашения ссуды нерегулярными платежами

Таблица 15.3. Формулы, используемые для вычисления параметров погашения ссуды нерегулярными платежами

Ячейка	Формула	Описание
D6	=ЕСЛИ (C6<>""; (C6-C5)/365*N5*ПС; "")	По этой формуле вычисляется размер выплаты по процентам на основании даты платежа
E6	=ЕСЛИ (C6<>""; B6-D6; "")	Чтобы получить размер основной выплаты, из суммы платежа вычитается сумма выплат по процентам
F6	=ЕСЛИ (C6<>""; F5+B6; "")	Эта формула прибавляет сумму данного платежа к общей сумме выплат
G6	=ЕСЛИ (C6<>""; G5+D6; "")	Эта формула прибавляет сумму выплат по процентам к общей сумме
H6	=ЕСЛИ (C6<>""; H5-E6; "")	Формула вычисляет остаток в конце каждого периода



Рабочую книгу с примерами использования таблиц данных с примером вычисления параметров погашения ссуды нерегулярными платежами можно найти в файле *irregular payments.xlsx* на веб-сайте книги. Локализованный файл называется *Нерегулярные платежи.xlsx*.

Расчет параметров инвестиций

При расчете параметров инвестиций основной интерес представляет вычисление прибыли от инвестиций с фиксированной процентной ставкой, таких как банковские сберегательные счета, депозиты или аннуитеты. Такого рода вычисления можно проводить для инвестиций, состоящих из одного или более вклада.



Рабочую книгу со всеми расчетами, приведенными в этом разделе, можно найти в файле `investment calculations.xlsx` на веб-сайте книги. Локализованный файл называется `Расчет инвестиций.xlsm`

Будущая стоимость одного вклада

Большинство инвестиций представляет собой одноразовый вклад на определенный срок, проценты по которому выплачиваются по окончании этого срока. В настоящем разделе описаны способы вычисления простых и сложных процентов.

Вычисление простого процента

В случае *простого процента* сумма выплаченных процентов не прибавляется к основной сумме вклада. Чтобы вычислить доход на основании простого процента, используется приведенная ниже формула.

Доход = Основная_Сумма * Процентная_Ставка * Срок

Например, предположим, что в банк на депозит вложено 10 000 руб. на один год, при условии, что простая годовая процентная ставка составляет 5%. Через год банк вернет сумму вклада, а также выплатит проценты в размере 500 руб. Итого сумма выплаты составит 10 500 руб. В этом случае, чтобы вычислить полученные проценты, следует умножить основную сумму (10 000 руб.) на процентную ставку (0,05) и на срок 1 год.

Если срок инвестиции меньше одного года, то простая процентная ставка будет соответствующим образом преобразована. Например, если в предыдущем примере срок вклада будет шесть месяцев, то по процентам банк выплатит только 250 руб., поскольку в этом случае годовая процентная ставка умножается на 6/12.

На рис. 15.10 показан рабочий лист, который содержит формулы, позволяющие вычислить простой процент. Формула в ячейке B7, приведенная ниже, вычисляет размер выплат по процентам в конце срока.

=B3*B4*B5

	A	B
1	Расчет выплат по простому проценту	
2		
3	Сумма инвестиции:	10 000,00 р.
4	Годовая процентная ставка:	5,00%
5	Срок (лет):	1
6		
7	Начислено по процентам:	500,00 р.
8	Сумма инвестиций в конце срока:	10 500,00 р.
	Простой	

Рис. 15.10. Вычисление простого процента

Формула в ячейке B8 суммирует размер выплат по процентам и размер инвестиции.

Вычисление сложного процента

	A	B	C
1	Расчет выплат по сложному проценту		
2	Ежемесячное начисление процентов		
3			
4	Сумма инвестиции:	10 000,00 р.	
5	Годовая процентная ставка:	5,00%	
6	Срок (мес.):	12	
7			
8	Месяц	Проценты	Баланс
9	Начальная сумма		10 000,00 р.
10	1	41,67 р.	10 041,67 р.
11	2	41,84 р.	10 083,51 р.
12	3	42,01 р.	10 125,52 р.
13	4	42,19 р.	10 167,71 р.
14	5	42,37 р.	10 210,08 р.
15	6	42,54 р.	10 252,62 р.
16	7	42,72 р.	10 295,34 р.
17	8	42,90 р.	10 338,24 р.
18	9	43,08 р.	10 381,31 р.
19	10	43,26 р.	10 424,57 р.
20	11	43,44 р.	10 468,00 р.
21	12	43,62 р.	10 511,62 р.
22			

Рис. 15.11. Чтобы вычислить сложный процент, можно создать последовательность формул

первый месяц). Полученная сумма снова добавляется к текущей сумме вклада. Чтобы вычислить объем инвестиции в конце срока, можно создать последовательность формул (рис. 15.11).

В столбце В содержатся формулы, вычисляющие выплаты по процентам для каждого месяца. Например, в ячейке В10 содержится следующая формула:

=C9*(\$B\$5*(1/12))

Формулы в столбце С суммируют начисления по процентам за текущий месяц с суммой инвестиции за предыдущий месяц. Например, в ячейке С10 содержится такая формула:

=C9+B10

	A	B
1	Расчет выплат по сложному проценту	
2	Вычисление с помощью одной формулы	
3		
4	Сумма инвестиции:	50 000,00 р.
5	Годовая процентная ставка:	5,75%
6	Число выплат в году	4
7	Срок (лет):	3
8		
9	Процентная ставка за период:	1,44%
10	Сумма инвестиций в конце срока:	59 340,67 р.
11	Начислено по процентам:	9 340,67 р.
12		
13	Сумма начислений (%):	6,23%
14		

Рис. 15.12. Вычисление сложного процента с помощью одной формулы

В большинстве случаев выплаты по процентам вычисляются на основе одного из типов сложного процента. Под *сложным процентом* понимается многократное начисление процентов, которые добавляются к общей сумме вклада, причем в следующий раз проценты начисляются как по основной сумме вклада, так и по начисленным ранее процентам.

Например, предположим, что сумма депозита в банке составляет 10 000 руб. Годовая процентная ставка составляет 5%, проценты начисляются каждый месяц. После первого месяца проценты начисляются по основной сумме вклада, а затем полученная сумма выплат по процентам добавляется к основной. В следующем месяце проценты будут начисляться уже по большей сумме (основная плюс выплаты по процентам за

В конце срока (12 месяцев) сумма инвестиции составляет 10 511,62 руб. Таким образом, ежемесячное начисление сложного процента принесло (по сравнению с простым процентом) дополнительную прибыль в размере 11,62 руб.

Вместо того чтобы создавать последовательность формул, можно воспользоваться функцией БС. На рис. 15.12 показан рабочий лист, который содержит формулы для вычисления сложного процента. В ячейке В6 хранится число выплат за один год. Если выплаты производятся ежемесячно, то в ячейке В6 должно находиться число 12. Если выплаты будут производиться ежеквартально, то число выплат

равно 4. Если же выплаты производятся ежедневно, то количество выплат будет равно 365. Срок инвестиции, указанный в виде количества лет, содержится в ячейке B7.

В ячейке B9 содержится приведенная ниже формула, которая вычисляет процентную ставку за один период. Полученное значение процентной ставки используется для вычисления суммы выплаты за каждый период.

=B5* (1/B6)

В формуле, которая хранится в ячейке B10, для расчета размера инвестиции в конце срока вклада используется функция БС:

=БС (B9;B6*B7;;-B4)

Первым аргументом функции БС является значение процентной ставки за один период, которое вычисляется в ячейке B9. Вторым аргументом представляет собой общее число периодов. Третий аргумент пропущен, а четвертый аргумент — это размер инвестиции, который записан как отрицательное число.

Общая сумма выплат по процентам вычисляется с помощью простой формулы, записанной в ячейке B11:

=B10-B4

Формула в ячейке B13 позволяет вычислить сумму начислений за год (выраженную в процентах) относительно объема инвестиции:

=(B11/B4) /B7

Например, пусть депозит в банке равен 50 000 руб., процентная ставка составляет 5,75% годовых, начисляемых ежеквартально. В этом случае количество периодов выплат за год равно 4 (ячейка B6). Срок вклада составляет три года, поэтому в ячейку B7 вводится число 3. Формула в ячейке B10 вернет значение 59 340,67 руб.

Предположим, что другой банк предлагает депозиты на тех же условиях, но начисления по процентам происходят ежедневно. На рис. 15.13 показаны параметры депозита размером 50 000 руб. с ежедневными выплатами. Сравните полученные значения со значениями на рис. 15.12. Как видно, разница очень мала (общая сумма выплат по процентам равна 9 340,67 и 9 412,78 руб. соответственно). За три года по второму вкладу будет выплачено на 72,11 руб. больше, чем по первому. Если же сравнивать суммы выплат за год, то при ежеквартальных начислениях эта сумма будет составлять 6,23%, а при ежедневных — 6,28%.

	A	B
1	Расчет выплат по сложному проценту	
2	Вычисление с помощью одной формулы	
3		
4	Сумма инвестиции:	50 000,00 р.
5	Годовая процентная ставка:	5,75%
6	Число выплат в году	365
7	Срок (лет):	3
8		
9	Процентная ставка за период:	0,02%
10	Сумма инвестиций в конце срока:	59 412,78 р.
11	Начислено по процентам:	9 412,78 р.
12		
13	Сумма начислений (%):	6,28%
14		
	« » Простой Сложный1 Сложный2	


Рис. 15.13. Вычисление параметров депозита с ежедневными выплатами

Вычисление сложного процента с постоянными начислениями

Термин *постоянные начисления* означает, что проценты начисляются и добавляются к сумме вклада непрерывно. Иными словами, число периодов выплат по инвестиции за год — бесконечно. Следующая формула позволяет определить будущую стоимость инвестиции объемом 50 000 руб., вложенную под 5,75% годовых, которые начисляются непрерывно на протяжении трех лет:

=50000*EXP (0,0575*3)

Эта формула вернет значение 59 413,59 руб., что на 0,08% больше по сравнению с аналогичной инвестицией, но с ежедневными выплатами.



Сложный процент можно вычислить без использования функции БС. Общая формула выглядит следующим образом:

$$\text{сумма} * (1 + \text{ставка}) ^ \text{кпер}$$

где *сумма* — основная сумма инвестиции; *ставка* — процентная ставка за один период; *кпер* — общее количество периодов выплат.

Например, пусть инвестиция размером 50 000 руб. вкладывается на пять лет под 5% годовых, которые начисляются ежемесячно. В этом случае будущую стоимость инвестиции можно вычислить по формуле

$$=50000 * (1 + 5\% / 12) ^ (12 * 5)$$

Правило числа 72

Что делать, если нужно быстро принять решение о том, вкладывать деньги на определенных условиях или нет, а компьютера поблизости нет? Простое “правило числа 72” позволяет определить количество лет, по истечении которых стоимость инвестиции, вложенной под определенные проценты с ежегодными начислениями, удвоится. Просто разделите число 72 на значение процентной ставки. Например, пусть 100 000 руб. вкладываются под 4% годовых. Через сколько лет эта сумма возрастет до 200 000 руб., т.е. удвоится? Разделив 72 на 4, получим, что стоимость данной инвестиции удвоится через 18 лет. А что будет, если процентная ставка будет равняться 5% годовых? В последнем случае сумма удвоится чуть больше, чем через 14 лет.

Теперь рассмотрим точность этого правила. В таблице, приведенной ниже, во втором столбце представлены значения, полученные с помощью “правила числа 72”, а в правом столбце — реальные значения. Как видно, для малых значений процентной ставки это простое правило обеспечивает удивительную точность. Тем не менее помните, что если процентная ставка будет равна 30% или больше, точность будет значительно меньше.

Процентная ставка	“Правило числа 72”	Фактическое значение
1%	72,00	69,66
2%	36,00	35,00
3%	24,00	23,45
4%	18,00	17,67
5%	14,40	14,21
6%	12,00	11,90
7%	10,29	10,24
8%	9,00	9,01
9%	8,00	8,04
10%	7,20	7,27
15%	4,80	4,96
20%	3,60	3,80
25%	2,88	3,11
30%	2,40	2,64

“Правило числа 72” работает и в обратном направлении. Например, чтобы удвоить стоимость инвестиции за шесть лет, следует разделить 72 на 6. В результате получим, что деньги нужно вкладывать под 12% годовых. Остается пожелать успехов!

Будущая стоимость серии вкладов

Рассмотрим еще один тип инвестиций, в котором на протяжении всего срока инвестиции следует периодически делать дополнительные вклады. Этот тип инвестиции называется *аннуитетом*. Некоторые банковские учреждения предлагают такую услугу.

Функции, описанные выше в разделе “Кредитный калькулятор”, также можно применить и для расчета аннуитетов, но для этого нужно поменять перспективу (кредитор вместо заемщика). Самый простой пример такого вида инвестиций — это целевые сберегательные вклады, предназначенные для проведения отпуска, предлагаемые целым рядом банковских организаций. Смысл состоит в том, что из каждой выплаты вам банк удерживает определенную фиксированную сумму, которая зачисляется в виде депозита на специальный сберегательный счет. Затем, через заранее оговоренный промежуток времени, например, один год, вы можете снять всю сумму (включая проценты) и использовать ее для проведения отпуска.

Предположим, что в начале каждого месяца (на протяжении 12 месяцев) на счет вносится 2 000 руб. под 4,25% годовых, начисляемых ежемесячно. Следующая формула позволяет вычислить будущую стоимость депозита:

$=\text{БС}(4,25\%/12;12;-2000;;1)$

Формула вернет значение 24 559,74 руб. Это число представляет собой сумму общего объема взносов (24 000 руб.) и начислений по процентам (559,74 руб.). Последний аргумент функции БС равен 1, это значит, что взносы производятся в начале каждого периода. На рис. 15.14 показан рабочий лист, который содержит формулы для расчета параметров аннуитета. В табл. 15.4 представлены формулы этого листа с краткими описаниями.

	A	B
1	Расчет параметров аннуитета	
2		
3	Депозиты...	
4	Первый взнос:	0,00 р.
5	Размер периодического взноса:	2 000,00 р.
6	Число взносов за год	12
7	Депозит вносится в начале периода?	ИСТИНА
8		
9	Сроки инвестиций...	
10	Срок инвестиции (лет):	1
11		
12	Процентная ставка...	
13	Годовая процентная ставка:	4,25%
14		
15	Расчеты...	
16	Первый взнос:	0,00 р.
17	Всего внесено:	24 000,00 р.
18	Всего внесено с учетом первого взноса:	24 000,00 р.
19	Процентная ставка за период:	0,35%
20	Сумма инвестиций в конце срока:	24 559,74 р.
21	Ваша прибыль:	559,74 р.
22		

Рис. 15.14. Расчет параметров аннуитета



Эту рабочую книгу можно найти в файле `annuity calculator.xlsx` на веб-сайте книги. Локализованный файл называется `Расчет аннуитета.xlsm`

Таблица 15.4. Формулы, используемые для вычисления параметров аннуитета

Ячейка	Формула	Описание
B4	Значение вводится	Размер первого взноса (может быть равным нулю)
B5	Значение вводится	Размер периодического взноса
B6	Значение вводится	Число взносов за год
B7	Значение вводится	ИСТИНА — если взносы вносятся в начале периода; ЛОЖЬ — в противном случае
B10	Значение вводится	Срок инвестиции (лет). Значение может быть дробным
B13	Значение вводится	Значение годовой процентной ставки

Окончание табл. 15.4

Ячейка	Формула	Описание
B16	=B4	Отображает размер первого взноса
B17	=B5*B6*B10	Вычисляет общую сумму сделанных взносов
B18	=B16+B17	Вычисляет сумму первого и всех сделанных взносов
B19	=B13*(1/B6)	Вычисляет процентную ставку за период
B20	=ЕС(B19; B6*B10; -B5; -B4; ЕСЛИ(B7;1;0))	Вычисляет будущую стоимость инвестиции
B21	=B20-B18	Вычисляет прибыль от инвестиции

Расчет амортизации

Программа Excel содержит пять функций, которые позволяют рассчитать амортизацию активов за время их эксплуатации. Значение амортизации в данный момент времени (уменьшение стоимости в заданный момент времени по сравнению с начальной стоимостью) можно описать с помощью функции, аргументами которой являются начальная стоимость и время эксплуатации. Выбор функции зависит от используемого *метода амортизации*.

В табл. 15.5 кратко описаны функции Excel, позволяющие рассчитать амортизацию, а также их аргументы. Дополнительную информацию можно найти в справочной системе Excel.

Таблица 15.5. Функции для расчета амортизации

Функция	Метод амортизации	Аргументы*
АПЛ	Линейный метод. Стоимость актива каждый год уменьшается на одинаковую сумму на протяжении всего срока эксплуатации	нач_стоимость; ост_стоимость; время_эксплуатации
ФУО	Метод фиксированного уменьшения остатка. Значение амортизации за период вычисляется с помощью фиксированной процентной ставки	нач_стоимость; ост_стоимость; время_эксплуатации; период; [месяцы]
ДДОВ	Метод двойного уменьшения остатка. Амортизация вычисляется с помощью увеличенного коэффициента. Значение амортизации максимально в первый период, а затем — постепенно уменьшается	нач_стоимость; ост_стоимость; время_эксплуатации; период; [коэффициент]
АСЧ	Метод суммы годовых чисел. Устанавливает большее значение амортизации в первые периоды эксплуатации	нач_стоимость; ост_стоимость; время_эксплуатации; период
ПУО	Метод снижающегося остатка. Вычисляет амортизацию для любого периода (включая частичные периоды) с использованием метода двойного уменьшения остатка или любого другого указанного метода	нач_стоимость; ост_стоимость; время_эксплуатации; нач_период; кон_период; [коэффициент]; [без_переключения]

*) Аргументы в квадратных скобках задавать необязательно.

Для того чтобы лучше понять принцип работы функций для расчета амортизации, рассмотрим их аргументы.

- **нач_стоимость** — начальная стоимость активов.
- **ост_стоимость** — стоимость активов в конце срока эксплуатации.

- **время_эксплуатации** — количество периодов, на протяжении которых эксплуатируются активы.
- **период** — период, по истечении которого производится расчет амортизации.
- **месяцы** — количество месяцев в первом году; если аргумент опущен, он полагается равным 12.
- **нач_период** — начальный период для вычисления амортизации.
- **кон_период** — конечный период для вычисления амортизации.
- **коэффициент** — процентная ставка снижающегося остатка; если аргумент опущен, то он полагается равным 2 (т.е. используется метод двойного уменьшения остатка).
- **без_переключения** — логическое значение, которое указывает, следует ли использовать линейную амортизацию, если значение амортизации превышает величину, рассчитанную методом снижающегося остатка.

На рис. 15.15 приведен пример расчета амортизации с помощью функций АПЛ, ФУО, ДДОБ и ФСЧ. Начальная стоимость актива составляет 100 000 руб., срок эксплуатации — 10 лет, остаточная стоимость — 10 000 руб. В диапазоне Значение амортизации представлены ежегодные объемы амортизации актива, рассчитанные разными методами. Во втором диапазоне Стоимость актива представлены значения уцененной стоимости актива на протяжении всего срока его эксплуатации.

	A	B	C	D	E
1	Актив:	Офисная мебель			
2	Начальная стоимость:	100 000,00 р.			
3	Срок эксл. (лет):	10			
4	Остаточная стоимость:	10 000,00 р.			
5					
6	Значение амортизации				
7	Год	АПЛ	ФУО	ДДОБ	АСЧ
8	1	9 000,00 р.	20 600,00 р.	20 000,00 р.	16 363,64 р.
9	2	9 000,00 р.	16 356,40 р.	16 000,00 р.	14 727,27 р.
10	3	9 000,00 р.	12 986,98 р.	12 800,00 р.	13 090,91 р.
11	4	9 000,00 р.	10 311,66 р.	10 240,00 р.	11 454,55 р.
12	5	9 000,00 р.	8 187,46 р.	8 192,00 р.	9 818,18 р.
13	6	9 000,00 р.	6 500,84 р.	6 553,60 р.	8 181,82 р.
14	7	9 000,00 р.	5 161,67 р.	5 242,88 р.	6 545,45 р.
15	8	9 000,00 р.	4 098,37 р.	4 194,30 р.	4 909,09 р.
16	9	9 000,00 р.	3 254,10 р.	3 355,44 р.	3 272,73 р.
17	10	9 000,00 р.	2 583,76 р.	2 684,35 р.	1 636,36 р.
18					
19					
20	Стоимость актива				
21	Год	АПЛ	ФУО	ДДОБ	АСЧ
22	0	100 000,00 р.	100 000,00 р.	100 000,00 р.	100 000,00 р.
23	1	91 000,00 р.	79 400,00 р.	80 000,00 р.	83 636,36 р.
24	2	82 000,00 р.	63 043,60 р.	64 000,00 р.	68 909,09 р.
25	3	73 000,00 р.	50 056,62 р.	51 200,00 р.	55 818,18 р.
26	4	64 000,00 р.	39 744,96 р.	40 960,00 р.	44 363,64 р.
27	5	55 000,00 р.	31 557,49 р.	32 768,00 р.	34 545,45 р.
28	6	46 000,00 р.	25 056,65 р.	26 214,40 р.	26 363,64 р.
29	7	37 000,00 р.	19 894,98 р.	20 971,52 р.	19 818,18 р.
30	8	28 000,00 р.	15 796,61 р.	16 777,22 р.	14 909,09 р.
31	9	19 000,00 р.	12 542,51 р.	13 421,77 р.	11 636,36 р.
32	10	10 000,00 р.	9 958,75 р.	10 737,42 р.	10 000,00 р.
33					

Рис. 15.15. Значения амортизации, полученные с использованием четырех функций Excel



Эту рабочую книгу можно найти в файле `depreciation calculations.xlsx` на веб-сайте книги. Локализованный файл называется `Расчет амортизации.xlsx`.

На рис. 15.16 представлен график остаточной стоимости активов, вычисленный с использованием различных методов амортизации. Как видно, с использованием функции АПЛ стоимость актива снижается по прямой линии. При использовании других функций стоимость снижается по кривой, поскольку максимальное снижение происходит в первые периоды срока эксплуатации.




Рис. 15.16. Расчет остаточной стоимости активов с помощью разных методов амортизации

	А	В
1	Актив:	Офисная мебель
2	Начальная стоимость:	100 000,00 р.
3	Срок эксл. (лет):	10
4	Остаточная стоимость:	10 000,00 р.
5		
6	Начальный период:	0
7	Конечный период:	3
8	Коэффициент:	2
9	Без переключения:	ИСТИНА
10		
11	Амортизация:	\$48 800,00
12		
	Амортизация	ПУО (+)

Рис. 15.17. Расчет амортизации за несколько периодов с помощью функции ПУО

Функция ПУО используется в тех случаях, когда нужно вычислить амортизацию за несколько периодов (например, за второй и третий годы эксплуатации). На рис. 15.17 показан рабочий лист, который позволяет выполнить расчет амортизации с использованием функции ПУО. В ячейке В11 содержится формула

`=ПУО(В2;В4;В3;В6;В7;В8;В9)`

которая рассчитывает амортизацию актива за первые три года эксплуатации (начиная с периода 0 и заканчивая периодом 3).